
INTRODUCTION

Zener macro--models provide an accurate SPICE
simulation of a diode’s current versus voltage characteristics.
Themacro--models arecreatedbycombining standardSPICE
devices into a sub--circuit. Zenermacro--models offer several
advantages over the standard diode model available in
SPICE, including a more accurate representation of the
breakdown characteristics.

Zener I vs. V Characteristics
The current versus voltage relationship of a Zener diode

is shown in Figure 1. The four operational regions are the
forward, leakage, breakdown and the maximum current
regions. Zener diodes are typically used as voltage
regulators; thus, the reverse bias breakdown region is the
normal operating area for the device.

on the data sheet)

Figure 1. Zener Diode I vs. V Characteristics and Specification Definitions

Zener Diode SPICE Model Options
The threebasicmethodsused to createZener diodeSPICE

models are listed below.
1. Curve Fit Models
2. SPICE Diode Statement
3. Macro--Model Subcircuit

Curve Fit Models
Curve fit SPICE models are an attractive modeling option

in versions of SPICE that have analog behavior modeling
statements. The mathematical functions and “If--Then--Else”
commands can be used to perform conditional branching to
accuratelymatch the Zener’s impedance in the four operating
regions. Curve fit models are not as popular as the Diode ‘D’
statement or macro--model options because the analog
behavioral statements are not available in some versions of
SPICE. In contrast, the SPICE statements used in the diode
and macro--model are compatible with almost all versions of
SPICE.
Curve fit models are created using a polynomial

expression in the SPICE voltage controlled current source
‘G’ command. The polynomials can be generated using the
curve fit feature of a mathematical program such as
MathCad or Excel to form a transfer function of the Zener’s
impedance. A negative feature of the curve fit approach is
that this option creates a mathematical model, while in
contrast themacro--model approach produces an equivalent
circuit to represent the Zener.

SPICE Diode Statement
The majority of the Zener SPICE models available in the

industry are created using the SPICE ‘D’ diode statement.
Table 1 provides the variables available with the PSPICE
‘D’ diode model. An example of a ‘D’ statement is shown
below.

* anode cathode model name
D1 2 1 MD1

.MODEL MD1 D IS=6.57933e--10 N=1.84949 XTI=1
+RS=1.022CJO=1.5e--10 TT=1e--08

There are several restrictions that limit the accuracy of
using the diode ‘D’ statement to model a Zener. First, the
diode statement does not have a provision for defining a
separate series resistance for the forward and reverse bias
breakdown regions. The resistances in the two regions are
not equal; thus, it is not possible to accurately model the
slope of the voltage versus current characteristic in both
regions. Next, the ‘D’ statement does not have a variable to
model the variance of the Zener voltage with temperature
(T). The polarity and magnitude of ΔVZ/ΔT is a function
of the breakdown voltage. If VZ is < 5.6 V, ΔVZ/ΔT ≈
--2.0 mV/_C. If VZ is > than 5.6 V, theΔVZ/ΔT temperature
coefficient is positive and increases as a function of the
magnitude of VZ.Also, the diode statement does not have a
variable to limit the current to a maximum value that
matches the Zener’s power dissipation capability.

Table 1. Default Values of the PSPICE Diode Statement ‘D’ Variables

Variable Parameter
SPICE Default

Value

ON
Macro--Model
Default Value Units

IS Saturation Current 1 E--14 1 E--14 A

RS Resistance 0 10 Ω

BV Reverse Breakdown Voltage ∞ 5 V

IBV Current at Reverse Breakdown Voltage 1 E--3 1 E--3 A

N Emission Coefficient (η) 1 1 --

XT1 Saturation Current Temp. Coefficient 3 0 --

TT Transit Time 0 10 nS

CJO Zero Bias Junction Capacitance 0 50 pF

VJ Junction Potential 1 1 V

M Grading Coefficient 0.5 0.5 --

EG Activation Energy 1.11 1.11 eV

KF Flicker Noise Coefficient 0 0 --

AF Flicker Noise Exponent 1 1 --

FC Depletion Capacitance Forward Bias Coefficient 0.5 0.5 --

TNOM Nominal Temperature 27 27 _C

1. The values of the ON macro--model components are generated by the MODPEX software from the data sheet specifications.
2. The PSPICE diode command does not have a maximum current limit (IMAX) variable. The macro--model’s default value for IZM is 200 mA.
3. The PSPICE diode command does not have a variable to define the temperature coefficient of the breakdown voltage. The macro--model’s

default value for ΔVZ/ΔT is 0 mV/_C.

Macro--Model Subcircuit
A Zener diode macro--model can be created using

standard SPICE components. Figure 2 shows a schematic
representation of the macro--model that is used to create the
majority of the ON Zener SPICEmodels. The netlist for the

MMBZ5232’s macro--model is provided in Appendix I as
an example. A summary of the operation of the model is
given in Table 2. Further details on the macro--model are
provided in references [5] and [6]. Alternative Zener diode
macro--models are given in references [1]--[4].

Figure 2. Zener Diode SPICE Macro--Model

Forward Bias Region
Diode D1 is the key component when the voltage VD

across the Zener is greater than zero. D1’s forward bias
characteristics are controlled by the saturation (IS),

emission coefficient (N) and series resistance (RS)
variables. The forward bias current equations are listed
below.

ID= IF+ IL+ IR

= IF_D1+
VD
RL
+ IS_D2 IL & IR≪ IF_D1

∴ID≅ IF_D1≅ IS_D1e ^ VD1ηVT
− 1 ≅ IS_D1e ^ VD1ηVT

VT= kT
q ≅ 26 mV @ 25˚C

Leakage Region
The leakage or reverse bias before breakdown region is

defined when voltage VD is between 0 V and the
breakdown voltage (VBR). Currents IF and IR are small in

comparison to IL because diodes D1 and D2 are reverse
biased. The leakage current can be approximated by the
ratio of VD to RL.

ID= IF+ IL+ IR

= IS_D1+
VD
RL
− IS_D2 IS_D1 & IS_D2≪ IL

∴ID≅
VD
RL

Breakdown Region
The reverse bias breakdown region is modeled by the

devices contained in the current path of IR. Current flows
through this path when voltage VD exceeds the voltage of
EV1 plus the forward voltage ofD2. The breakdown voltage

(VBR), represented byEV1, is equal to the product of current
source IBV and resistor RBV. The Zener voltage VZ,
specified at current IZT, is equal to the sum of the voltages of
EV1, RZ, D2 and D3 as shown below.

ID≅ ISe ^  VDηVT ∴VD≅ ηVTInIDIS
VZ= VBR+ VD2+ VRZ− VD3

VZ @ IZT= EV1+ η2VT lnID2IS2
+ (IZTRZ)− η3VT lnID3IS3


= (IBRRBV)+ η2VT lnIZTIS2+ (IZTRZ)− η3VT lnIZG--IZTIS3

An equation that determines the ratio of the emission
coefficientsη2 andη3 is created by assuming that the reverse
saturation currents for diodes D2 and D3 are equal.

η3= η2 lnIZG--IZT
IZT


The Zener impedance, or the slope of the current versus
voltage curve, is equal to the derivative of the diode’s
voltage equation with respect to current. The impedance
equation shown below is valid for low frequencies, where
the junction capacitance can be neglected.

ZD=
δVD
δID

=
ηVT
ID

The impedance of the Zener, defined as ZZT and ZZK, is
defined at test currents IZT and IZK, respectively. The two
impedance and η3 equations are used to calculate RZ, η2 and
η3. The Zener impedance in the breakdown region is
approximated by neglecting the parallel connected resistor
RL and the reverse biased diode D1 for simplicity.

ZZ≅ RZ+ ZD2+ ZD3

ZZT≅ RZ+
η2VT
IZT

+
η3VT

(IZG--IZT)

ZZK≅ RZ+
η2VT
IZK

+
η3VT

(IZG--IZK)

Current Limit Region
Diode D3, current source IZG and resistor RZG are used to

form a current limiting circuit. The current limiting mode
occurswhen ID2 increases to the value of IZG,which reverse
biases D3 and limits the current. The maximum current
(IZM) has a weak dependency on the magnitude of the bias
voltage and RZG is used to provide a slope to the current
versus voltage curve. D3’s current can be expressed by the
equations listed below.

ID3= IZG− ID2− IRZG
IRZG≪ IZG & ID2

∴ID3≅ IZG− ID2

Macro--Model Limitations
Macro--models provide an accurate SPICE representation

of a Zener’s current and voltage characteristics for most
applications. The macro--models solve several of the
limitations associated with the SPICE diode ‘D’ statement
and the curve fitmodels. The accuracy of themacro--models
is directly proportional to the thoroughness of the Zener’s
data sheet. Macro--models are a powerful analytical design
tool; however, they should not be used as a replacement for
hardware development tests. A summary of the limitations
of the macro--models is shown in Table 3.

Table 2. Key Design Equations and Features of the Zener Diode Macro--Model

Region
Voltage

Boundaries Key Design Equations
Macro--Model
Components Comments

Forward
Bias

VD > 0

ID≅ IF_D1≅ IS_D1e ^ VD1ηVT
 D1 • ID is defined by the Ebers--Moll

equation

• D1’s RS variable models the ΔI/ΔV
slope

Leakage
Region

VBR < VD ≤ 0
ID≅ IL≅

VD
RL

RL • RL’s temperature coefficients model
ΔIL/ΔT

Breakdown VZM < VD ≤ VBR VZ≅ VZT+ ZZTID
VZT= VZ @ IZT
ZZT≅ RZ+ RSD2+ RSD3

EV1, D2, RZ
IBV, RBV

• The recommended bias current for a
voltage regulator application is IZM < ID
≤ IZT

• RBV’s temperature coefficients model
ΔVZ/ΔT

• RZ models the ΔI/ΔV slope

Current
Limit

VD ≤ VZM IZM≅ IZG D3, IZG, RZG • IZM is a function of the IC package

• RZG models the ΔI/ΔV slope

Table 3. Simulation Limits of Zener Macro--Models

Region Key Design Parameter Model Limitation

Forward Forward Voltage (VF) • VF is typically specified as a maximum value at a single current point in the data
sheet

• The simulation accuracy is enhanced if two typical current points are used by the
modeling algorithm

Leakage Leakage Current (IL) • IL is modeled as a linear function of the bias voltage

• IL actually varies as an exponential function of the bias voltage

Breakdown Zener Voltage (VZ) • VZ tolerance (typ.5%) is not modeled
• Monte--Carlo simulations can provide tolerance and worst case analysis

Current Limit Maximum Current (IMAX) • Voltage surge suppression capability beyond IMAX is not modeled

• Thermal self--heating produced when ID is large is not modeled

• Device overcurrent failures are not modeled. At the device’s destruction point, VZ
‘collapses’ or decreases to a low value, which increases the current through the
device to a level that damages the device

Appendix I: MMBZ5232B Zener Diode Macro--Model SPICE Netlist
**********************************************************************************
* MMBZ5232B Macro--Model SPICE Model
* Model Generated by MODPEX
* Copyright(c) Symmetry Design Systems
* All Rights Reserved
* MODEL FORMAT: PSPICE
****************************************************************************
* node: anode cathode
.SUBCKT mmbz5232blt1 2 1
**********************************************************************************
* Forward Region
* D1’s CJO term models the Zener’s capacitance
D1 2 1 MD1
.MODEL MD1 D IS=7.58703e--07 N=3.10159 XTI=1 RS=0.856
+ CJO=1.5e--10 TT=1e--08
****************************************************************************
* Leakage Region
* RL models leakage current (IL)
* MDR temp. coef. model ∆IL / ∆T
RL 1 2 MDR 1.5e+09
.MODEL MDR RES TC1=0 TC2=0
****************************************************************************
* Reverse Breakdown Region
* RZ models the ∆I / ∆V slope
RZ 2 3 0.862776
D2 5 4 MD2
.MODEL MD2 D IS=2.5e--12 N=0.475376 XTI=0 EG=0.1
EV1 1 5 6 0 1
* Breakdown Voltage (VBR) = IBV x RBV
* MDRBV temp. coef. model ∆VBR / ∆T
IBV 0 6 0.001
RBV 6 0 MDRBV 5431.8
.MODEL MDRBV RES TC1=0.000368055
****************************************************************************
* Current Limit Region
* Maximum current (IZM) is set to magnitude of IZG
* RZG models the ∆I / ∆V slope
IZG 4 3 0.24
RZG 4 3 75
D3 3 4 MD3
.MODEL MD3 D IS=2.5e--12 N=0.198247 XTI=0 EG=0.1
*
.ENDS mmbz5232blt1
****************************************************************************

